
'Program Notes:

'                                                                               A D O  I S S U E S

'This app uses ADO 2.0 to export data from my form to MS-EXCEL! The code is not any different than the code

'needed to add a record to a database. Look at the connection string. It tells VB we are using Excel as out data

'store and not SQL or Access or Oracle.

'In order to simplify the process of adding records (rows) to Excel, I use parameters in my SQL statement. Look

'at the INSERT INTO statement. Notice I use the column names (the text in the top cells in Excel) and then I 

'follow that with a Values (   )  section. This is where you put in the data. I could use something like this:

'                 "Insert into [Sheet1$] (FirstName, LastName) values ('Bill', 'Brown')"

'Notice how I put the literal string data in single quotes '  '. Use this if you know what the data is or what

'you want it to be.

'If you don't know, or you want to grad data from a textbox, then the SQl statement gets real messy and I

'have a terrible time creating them. So I use another option. I tell my Command Object to equate a placeholder (parameter)

'to the textbox or variable. Then it will grab the data from the correct place and I don't have to worry about

'formatting sql strings!

'                                                                         O T H E R   F E A T U R E S

'The MS logo is inside a picture box. When the mouse hovers over the logo the icon turns to a hand. When they click on the

'logo, it launches the browser and goes to the MS Excel website!

Imports System.Data

Imports System.Data.OleDb

Public Class frmMain

    Private Sub btnAdd_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnAdd.Click

        '---define and grab data from boxes

        Dim firstName As String = Trim(txtFirst.Text)

        Dim lastName As String = Trim(txtLast.Text)

        Dim telephone As String = Trim(txtTele.Text)

        '---let's make sure they typed something!!!!

        If firstName = "" Or lastName = "" Or telephone = "" Then

            MessageBox.Show("Please enter all data", "System Message", MessageBoxButtons.OK, MessageBoxIcon.Warning)

            txtFirst.Focus()

        Else

            'Establish a connection to the data source. By declaring it outside the Try/Catch stuff, you can refer to conn later

            'and be able to close it....you will see.

            Dim connString As String

            connString = "Provider=Microsoft.Jet.OLEDB.4.0;Data Source= ..\..\data\customers.xls;Extended Properties=Excel 8.0;"

            '---this creates a connection object...this is NOT the same as the connString above

            Dim conn As New OleDbConnection(connString)

            Try

                'the @ symbol below is used to create a placeholder/parameter. It tells SQL the value will

                'be created/explained/dealt with later in a Parameters statement!

                Dim insertString As String

                insertString = "Insert into [Sheet1$] (FirstName, LastName, Telephone) Values (@FName, @LName, @Tele)"

                Dim dbCommandObject As New System.Data.OleDb.OleDbCommand(insertString, conn)

                '---addwithValue is .net 2005 verision. Here I tell Mr. Command Object that @FName = firstName, etc.

                dbCommandObject.Parameters.AddWithValue("@FName", firstName)

                dbCommandObject.Parameters.AddWithValue("@LName", lastName)

                dbCommandObject.Parameters.AddWithValue("@Tele", telephone)

                conn.Open()   'keep connections open as short of a time as possible!

                dbCommandObject.ExecuteNonQuery()     'this updates a data store w/o needing a dataset!

                conn.Close()

                '---give thema nice feeling here!

                MessageBox.Show("Your data was exported to Excel.", "Operation Complete", MessageBoxButtons.OK, MessageBoxIcon.Information)

                txtFirst.Clear()

                txtLast.Clear()

                txtTele.Clear()

                txtFirst.Focus()

            Catch ex As Exception      'I am catching all errors/exceptions and appending the description to my message. Err is called the Error object.

                MessageBox.Show("I had a problem exporting your data. " & Err.Description, "System Message", MessageBoxButtons.OK, MessageBoxIcon.Error)

                Me.Close()

            Finally

                'Closes the connection whether an error occurs or not. This is just good practice!

                conn.Close()

            End Try

        End If

    End Sub

    Private Sub btnEnd_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnEnd.Click

        Me.Close()

    End Sub

    Private Sub PictureBox1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles PictureBox1.Click

        System.Diagnostics.Process.Start("http://www.microsoft.com/office")

    End Sub

    Private Sub PictureBox1_MouseHover(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles PictureBox1.MouseHover

        PictureBox1.Cursor = Cursors.Hand

    End Sub

End Class

